12.3 Table Look-up With Hermite Cubic Interpolation

A. Purpose

Given a table of values of a function and its first derivative, this subroutine does table look-up and Hermite cubic interpolation to compute an interpolated value of the function for an arbitrary argument. Optionally the subroutine can compute values of the first, second, or third derivative of the interpolation function. All features here are also available, with more generality, in Chapter 12.1.

The interpolated function will match the given data at the tabular points, and thus will have a continuous value and first derivative. In general the second and third derivatives will have jump discontinuities at the tabular points. It is possible, of course, for the data to satisfy relations that give rise to continuity of second or third derivatives in the interpolated function. In particular, data from the cubic spline fitting programs, DC2FIT or SC2FIT of Chapter 11.4, will produce continuous second derivatives.

B. Usage

B.1 Program Prototype, Single Precision

INTEGER NDERIV, NTAB

REAL SHINT, X, XTAB(≥NTAB)**,**
YTAB(≥ NTAB)**, YPTAB**(≥NTAB)**, YVAL**

Assign values to X, NDERIV, NTAB, XTAB(), YTAB(), and YPTAB(). In particular, NTAB, XTAB(), YTAB(), and YPTAB() may have the same values as on return from a previous call to SC2FIT.

YVAL = SHINT(X, NDERIV, NTAB, XTAB, YTAB, YPTAB)

This reference will set YVAL to the value (or a derivative value as selected by NDERIV) of the curve defined by NTAB, XTAB(), YTAB(), and YPTAB().

B.2 Argument Definitions

X [in] Abscissa at which interpolation is to be done. If X is outside the domain of the table given in XTAB(), extrapolation will be done using the two tabular points at the nearest end of the table.

NDERIV [in] Set by the user to 0, 1, 2, or 3 to select computation of the value, or the first, second, or third derivative of the interpolation function.

NTAB [in] Number of values in each of the arrays, XTAB(), YTAB(), and YPTAB(). Require NTAB ≥ 2.

XTAB() [in] Abscissae of the given data. These values must be either strictly increasing or strictly decreasing.

YTAB(), YPTAB() [in] Values and first derivatives associated with the abscissae given in XTAB().

$$\mathrm{YTAB}(i) = f(\mathrm{XTAB}(i)),\ i = 1, ..., \mathrm{NTAB}$$
$$\mathrm{YPTAB}(i) = f'(\mathrm{XTAB}(i)),\ i = 1, ..., \mathrm{NTAB}$$

SHINT [out] Returns the value or selected derivative value computed by interpolation.

B.3 Modifications for double precision

For double-precision usage change the REAL type statements to DOUBLE PRECISION and change the function name from SHINT to DHINT. Note particularly that since this is a function subprogram the name DHINT must be typed either explicitly or via an IMPLICIT statement if it is used.

C. Examples and Remarks

The program DRSHINT illustrates the use of SHINT. This program first builds a table of values of the exponential function and its first derivative at three points, XTAB() = 0, 0.5, and 0.75. The program DRSHINT then uses SHINT to interpolate for the value and first derivative at a set of 21 points, $x = 0$ to 1.0 in steps of 0.05. Output is shown in ODSHINT. The first two columns show x and $\exp(x)$. Columns headed YINT and YPINT contain the interpolated value and the interpolated first derivative, respectively. The exponential function was chosen for convenience in this demonstration since at any point the true function value and all derivatives have the same value. Note that the function values are approximated more accurately than are the derivative values. Also note that the approximations are better on the interval from 0.5 to 0.75 than on the longer interval from 0 to 0.5. For $x > 0.75$ extrapolation is taking place, leading to much larger errors.

Interpolation error bounds are given in Section D. Applying these bounds to this example, ignoring the $O(h^4)$ term in the bound for the first derivative error, the error bounds are:

for f in $[0, 0.5] = (0.00260)(1.65)(0.5)^4 = 0.000268$
for f in $[0.5, 0.75] = (0.00260)(2.12)(0.25)^4 = 0.000022$
for f' in $[0, 0.5] = (0.00802)(1.65)(0.5)^3 = 0.00165$
for f' in $[0.5, 0.75] = (0.00802)(2.12)(0.25)^3 = 0.000266$

Note that these bounds are consistent with the results shown in ODSHINT.

D. Functional Description

Hermite interpolation together with the error in such interpolation is described in [1, pp. 314–317].

SHINT maintains a saved variable, LOOK, with an initial value of 1. The look-up procedure always starts with the tabular interval indexed by the saved value of LOOK, unless the saved value is not in [1, NTAB − 1]. The look-up procedure is a linear search, either forward or backward, from the starting value of LOOK. This is efficient for cases in which a succession of look-ups are done for successive arguments that are not far apart relative to the tabular spacing.

Each interpolation or extrapolation uses function values and first derivative values at two adjacent tabular points and is exact for cubic polynomial data. If the interpolation abscissa, x, is equal to an interior tabular abscissa, say x_i, then tabular data associated with x_{i-1} and x_i will be used for the computation. Extrapolation uses the first or last two tabular points, as appropriate.

If the given data are samples from a function having at least a fourth order derivative, and this derivative is bounded in magnitude by M_4 over the interval $[x_i, x_{i+1}]$, of length, h, then the error, E_i, in computing the interpolated derivative of order i for any argument in this interval is bounded as follows:

$$
\begin{aligned}
|E_0| &\le c_0 M_4 h^4 k, & c_0 &= 1/384 &&\approx 0.00260\\
|E_1| &\le c_1 M_4 h^3 + O(h^4), & c_1 &= \sqrt{3}/216 &&\approx 0.00802\\
|E_2| &\le c_2 M_4 h^2 + O(h^3), & c_2 &= 1/12 &&\approx 0.0833\\
|E_3| &\le c_3 M_4 h + O(h^2), & c_3 &= 0.5
\end{aligned}
$$

References

1. F. B. Hildebrand, **Introduction to Numerical Analysis**, McGraw-Hill, New York (1956) 511 pages. A reprint of a later edition is available from Dover.

E. Error Procedures and Restrictions

This subprogram requires NTAB $\ge$ 2 and 0 $\le$ NDERIV $\le$ 3. The values in the XTAB() array must either be strictly increasing or strictly decreasing. These conditions are not checked. Violation of these conditions will cause unpredictable effects.

F. Supporting Information

The source language is ANSI Fortran 77.

Entry	Required Files
DHINT	DHINT
SHINT	SHINT

Original code designed by C. L. Lawson and R. J. Hanson, JPL, 1968. Programmed and modified by Lawson, Hanson, T. Lang, and D. Campbell, JPL, 1968–1974. Adapted to Fortran 77 by Lawson and S. Y. Chiu, July 1987.

DRSHINT

```
c     program DRSHINT
c>> 1996-05-28 DRSHINT  Krogh Added external statement.
c>> 1994-10-19 DRSHINT  Krogh  Changes to use M77CON
c>> 1987-12-09 DRSHINT  Lawson  Initial Code.
c--S replaces "?": DR?HINT, ?HINT
c
      integer I, J, NTAB
      real        X, XTAB(3), YTAB(3), YPTAB(3), YINT, YPINT, YTRUE
      external SHINT
      real        SHINT
c
      data XTAB / 0.E0, .5E0, .75E0 /
      data NTAB / 3 /
c
   10 format(1X,'Demonstration of SHINT by interpolating to the',
     *         ' exponential function,'/
     *         1X,'given values at X = 0.0, 0.5, and 0.75'//
     *         4X,'X',7X,'YTRUE',9X,
     *         'YINT',6X,'YINT-YTRUE',5X,'YPINT',6X,'YPINT-YTRUE'/1X)
   20 format(1X,F5.2,5F13.6)
```

```
c
      do 30 I = 1,3
        YTAB(I) = EXP(XTAB(I))
        YPTAB(I) = YTAB(I)
   30 continue
c
      print 10
c
      do 50 J = 10, 110, 5
        X = REAL(J)/100.E0 - 0.1E0
        YINT = SHINT(X,0,NTAB,XTAB,YTAB,YPTAB)
        YPINT = SHINT(X,1,NTAB,XTAB,YTAB,YPTAB)
        YTRUE = EXP(X)
        print 20, X, YTRUE, YINT, YINT-YTRUE, YPINT, YPINT-YTRUE
   50 continue
c
      end
```

ODSHINT

```
Demonstration of SHINT by interpolating to the exponential function,
given values at X = 0.0, 0.5, and 0.75
```

X	YTRUE	YINT	YINT–YTRUE	YPINT	YPINT–YTRUE
0.00	1.000000	1.000000	0.000000	1.000000	0.000000
0.05	1.051271	1.051245	−0.000026	1.050336	−0.000935
0.10	1.105171	1.105088	−0.000083	1.103903	−0.001268
0.15	1.161834	1.161689	−0.000145	1.160699	−0.001135
0.20	1.221403	1.221211	−0.000192	1.220726	−0.000677
0.25	1.284025	1.283816	−0.000210	1.283983	−0.000042
0.30	1.349859	1.349663	−0.000196	1.350471	0.000612
0.35	1.419068	1.418916	−0.000151	1.420188	0.001120
0.40	1.491825	1.491736	−0.000089	1.493136	0.001311
0.45	1.568312	1.568284	−0.000028	1.569313	0.001001
0.50	1.648721	1.648721	0.000000	1.648721	0.000000
0.55	1.733253	1.733245	−0.000008	1.733022	−0.000231
0.60	1.822119	1.822101	−0.000018	1.822000	−0.000119
0.65	1.915541	1.915523	−0.000018	1.915656	0.000115
0.70	2.013753	2.013745	−0.000008	2.013989	0.000236
0.75	2.117000	2.117000	0.000000	2.117000	0.000000
0.80	2.225541	2.225523	−0.000018	2.224689	−0.000852
0.85	2.339647	2.339547	−0.000100	2.337056	−0.002591
0.90	2.459603	2.459306	−0.000297	2.454099	−0.005504
0.95	2.585710	2.585035	−0.000674	2.575821	−0.009889
1.00	2.718282	2.716967	−0.001315	2.702221	−0.016061